P 10

PRACTICE 10: VECTOR FIELDS

```
Clear["Global`*"]
```

▼ Proposed exercise P-10.1

Given the following vector field:

$$\vec{F}(x,y)=x*y\,\vec{i}+\left(x^2-y^2\right)\vec{j},$$

calculate the current lines and plot them together with the vector field.

▼ Resolution P-10.1

★ Differential Equation of the current lines:

The tangent vector of the curve in each point is given by (m(x,y),n(x,y)) and this is the vector field, being its slope n(x,y)/m(x,y)

```
m[x_, y_] = x * y;
n[x_, y_] = x^2 - y^2;
```

Definition and resolution of the Differential Equation of the current lines

```
edlc = y'[x] == n[x, y[x]] / m[x, y[x]]
```

$$y'[x] == \frac{x^2 - y[x]^2}{x\, y[x]}$$

```
S = DSolve[edlc, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{\sqrt{x^4 + 2\,C[1]}}{\sqrt{2}\,x}\right\}, \left\{y[x] \to \frac{\sqrt{x^4 + 2\,C[1]}}{\sqrt{2}\,x}\right\}\right\}$$

Taking into account the results obtained, two functions that are solutions of the Differential Equation are defined

```
s1[x_, c_] = S[[1, 1, 2]] /. C[1] → c
s2[x_, c_] = S[[2, 1, 2]] /. C[1] → c
```

$$-\frac{\sqrt{2\,c + x^4}}{\sqrt{2}\,x}$$

$$\frac{\sqrt{2\,c + x^4}}{\sqrt{2}\,x}$$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

Using the following list, a family of solutions is obtained. Its graphical representation is done

```
sol = Flatten[Table[{s1[x, c], s2[x, c]}, {c, 0.1, 10, 1}], 2]
```

$$\left\{-\frac{\sqrt{0.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{0.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{2.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{2.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{4.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{4.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{6.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{6.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{8.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{8.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{10.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{10.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{12.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{12.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{14.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{14.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{16.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{16.2+x^4}}{\sqrt{2}\,x}, -\frac{\sqrt{18.2+x^4}}{\sqrt{2}\,x}, \frac{\sqrt{18.2+x^4}}{\sqrt{2}\,x}\right\}$$

```
clines = Plot[Evaluate[sol], {x, -3, 3}, PlotRange → {-4, 4}]
```



The tangent vector of the curve in each point is given by (m(x,y),n(x,y)) and this is the vector field

```
vecf = VectorPlot[{m[x, y], n[x, y]}, {x, -3, 3}, {y, -4, 4}, Axes → True];
```


E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

★ Plot of the vector field and the current lines

```
Show[{vecf, clines}, PlotRange → {-4, 4}]
```



▼ Proposed Exercise P-10.2

Given the following vector field:

$$\vec{F}(x,y)=y\,\vec{i}-x\,\vec{j}$$

a) Calculate its vector field and plot it.

b) Solve the Differential Equation of the current lines and obtain the general solution.

c) Calculate and plot a family of solutions.

d) Plot in the same figure the family of curves and the vector field.

▼ Resolution P-10.2

★ Section a)

```
m[x_, y_] = y;
n[x_, y_] = -x;
```

```
vf = VectorPlot[{m[x, y], n[x, y]}, {x, -1, 1}, {y, -1, 1}, Axes → True]
```



★ Section b)

```
edlc = n[x, y[x]] / m[x, y[x]] == y'[x]
```

$$-\frac{x}{y[x]} == y'[x]$$

```
S = DSolve[edlc, y[x], x]
```

$$\left\{\left\{y[x] \to -\sqrt{-x^2 + 2\,C[1]}\right\}, \left\{y[x] \to \sqrt{-x^2 + 2\,C[1]}\right\}\right\}$$

```
s1[x_, c_] = S[[1, 1, 2]] /. C[1] → c / 2
s2[x_, c_] = S[[2, 1, 2]] /. C[1] → c / 2
```

$$-\sqrt{c - x^2}$$

$$\sqrt{c - x^2}$$

★ Section c)

```
sol = Flatten[Table[{s1[x, c], s2[x, c]}, {c, 0.01, 1, .1}], 2]
```

$$\left\{-\sqrt{0.01 - x^2}, \sqrt{0.01 - x^2}, -\sqrt{0.11 - x^2}, \sqrt{0.11 - x^2}, -\sqrt{0.21 - x^2}, \sqrt{0.21 - x^2}, -\sqrt{0.31 - x^2}, \sqrt{0.31 - x^2}, -\sqrt{0.41 - x^2}, \sqrt{0.41 - x^2}, -\sqrt{0.51 - x^2}, \sqrt{0.51 - x^2}, -\sqrt{0.61 - x^2}, \sqrt{0.61 - x^2}, -\sqrt{0.71 - x^2}, \sqrt{0.71 - x^2}, -\sqrt{0.81 - x^2}, \sqrt{0.81 - x^2}, -\sqrt{0.91 - x^2}, \sqrt{0.91 - x^2}\right\}$$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
familysol = Plot[Evaluate[sol], {x, -1, 1}, AspectRatio → Automatic]
```



★ Section d)

```
Show[{vf, familysol}]
```



▼ Proposed Exercise P-10.3

Given the following vector field $\vec{F}(x,y)=x\,\vec{i}+2\,y\,\vec{j}$

a) Calculate its vector field and plot it.

b) Solve the Differential Equation of the current lines and obtain the general solution.

c) Plot in the same figure the family of curves and the vector field.

d) Calculate the Differential Equation of the orthogonal trajectories.

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

e) Plot in the same figure the family of curbes and the vector field of the orthogonal trajectories.

f) Plot in the same figure both family of curves and the vector fields of each pf them.

▼ Resolution P-10.3

★ Section a)

```
m[x_, y_] = x;
n[x_, y_] = 2 y;
```

```
vf1 = VectorPlot[{m[x, y], n[x, y]}, {x, -1.5, 1.5}, {y, -1, 1}, Axes → True]
```



★ Section b)

```
edlc = n[x, y[x]] / m[x, y[x]] == y'[x]
```

$$\frac{2\,y[x]}{x} == y'[x]$$

```
Si = DSolve[edlc, y[x], x]
```

$\{\{y[x] \to x^2\, C[1]\}\}$

```
si[x_, c_] = Si[[1, 1, 2]] /. C[1] → c
```

$c\, x^2$

★ Section c)

```
solti = Flatten[{Table[si[x, c], {c, -0.5, 0.5, .1}], Table[si[x, c], {c, -10, 10, 2}]}, 2]
```

$\{-0.5\,x^2, -0.4\,x^2, -0.3\,x^2, -0.2\,x^2, -0.1\,x^2, 0., 0.1\,x^2, 0.2\,x^2, 0.3\,x^2, 0.4\,x^2, 0.5\,x^2, -10\,x^2, -8\,x^2, -6\,x^2, -4\,x^2, -2\,x^2, 0, 2\,x^2, 4\,x^2, 6\,x^2, 8\,x^2, 10\,x^2\}$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
famsolti = Plot[Evaluate[solti], {x, -1.5, 1.5},
  PlotRange → {-1, 1}, AspectRatio → Automatic, Ticks → None]
```



```
Show[{vf1, famsolti}]
```



★ Section d)

```
edto = y'[x] == -m[x, y[x]] / n[x, y[x]]
```

$$y'[x] == -\frac{x}{2\,y[x]}$$

```
Sto = DSolve[edto, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{\sqrt{-x^2 + 4\,C[1]}}{\sqrt{2}}\right\}, \left\{y[x] \to \frac{\sqrt{-x^2 + 4\,C[1]}}{\sqrt{2}}\right\}\right\}$$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
sto1[x_, c_] = Sto[[1, 1, 2]] /. C[1] → c / 4
sto2[x_, c_] = Sto[[2, 1, 2]] /. C[1] → c / 4
```

$$-\frac{\sqrt{c-x^2}}{\sqrt{2}}$$

$$\frac{\sqrt{c-x^2}}{\sqrt{2}}$$

★ Section e)

```
solto = Flatten[Table[{sto1[x, c], sto2[x, c]}, {c, 0.2, 2, .2}], 2]
```

$$\left\{-\frac{\sqrt{0.2-x^2}}{\sqrt{2}}, \frac{\sqrt{0.2-x^2}}{\sqrt{2}}, -\frac{\sqrt{0.4-x^2}}{\sqrt{2}}, \frac{\sqrt{0.4-x^2}}{\sqrt{2}}, -\frac{\sqrt{0.6-x^2}}{\sqrt{2}}, \frac{\sqrt{0.6-x^2}}{\sqrt{2}}, -\frac{\sqrt{0.8-x^2}}{\sqrt{2}}, \frac{\sqrt{0.8-x^2}}{\sqrt{2}}, -\frac{\sqrt{1.-x^2}}{\sqrt{2}}, \frac{\sqrt{1.-x^2}}{\sqrt{2}}, -\frac{\sqrt{1.2-x^2}}{\sqrt{2}}, \frac{\sqrt{1.2-x^2}}{\sqrt{2}}, -\frac{\sqrt{1.4-x^2}}{\sqrt{2}}, \frac{\sqrt{1.4-x^2}}{\sqrt{2}}, -\frac{\sqrt{1.6-x^2}}{\sqrt{2}}, \frac{\sqrt{1.6-x^2}}{\sqrt{2}}, -\frac{\sqrt{1.8-x^2}}{\sqrt{2}}, \frac{\sqrt{1.8-x^2}}{\sqrt{2}}, -\frac{\sqrt{2.-x^2}}{\sqrt{2}}, \frac{\sqrt{2.-x^2}}{\sqrt{2}}\right\}$$

```
famsolto = Plot[Evaluate[solto], {x, -1.5, 1.5}, PlotRange → {-1, 1}]
```




```
vfto = VectorPlot[{2 y, -x}, {x, -1.5, 1.5}, {y, -1, 1}, AspectRatio → Automatic]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Show[{vfto, famsolto}, PlotRange → {-1, 1}, Ticks → None, AspectRatio → Automatic]
```



★ Section f)

```
Show[{famsolto, famsolti}, PlotRange → {-1, 1}, Ticks → None]
```



```
Show[{vfto, vf1}, PlotRange → {-1, 1}, AspectRatio → Automatic]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Show[{vfto, vf1, famsolto, famsolti}, PlotRange → {-1, 1}, AspectRatio → Automatic]
```